TOM matching in Java

tom.loria.fr

List handling

List representation

Usually we consider an empty list nil and a concatenation operator cons

$$nil : \rightarrow L$$

$$cons : E \times L \rightarrow L$$

The list a -> b -> c is represented by

cons(a,cons(b,cons(c,nil)))

Question

- How to retrieve a given element?

in(cons(a,cons(b,cons(c,nil))),a) → true

in(cons(a,cons(b,cons(c,nil))),b) → true

in(cons(a,cons(b,cons(c,nil))),d) → false

Possible solution

in(cons(x,l),x) → true

in(cons(y,l),x) → in(l,x)

in(nil,x) → false

Associative symbols

- a symbol f is associative if

$$f(x,f(y,z)) = f(f(x,y),z)$$

- which one of these statements is true?
  - f(a,f(b,f(c,d))) = f(f(a,b),f(c,d)) ?
  - σ such that σf(x,f(c,y)) = f(f(a,b),f(c,d)) ?
  - f(x,f(c,y)) << f(f(a,b),f(c,d)) ?
  - f(x,f(c,y)) << f(a,f(b,f(c,d))) ?
  - f(x,f(d,y)) << f(a,f(b,f(c,d))) ?

Associativity with neutral element

- a symbol f is associative with neutral element e if

$$f(x,f(y,z)) = f(f(x,y),z)$$

$$f(x,e) = f(e,x) = x$$

- In this case do we have
  - f(x,f(c,y)) << f(a,f(b,f(c,d))) ?
  - f(x,f(d,y)) << f(a,f(b,f(c,d))) ?

Associative symbols in Tom

- A variadic symbol f with profile:

  f : E x … x E -> L

- is denoted

  f(E*) -> L

- and we can write

  f(a(),b(),c(),d())

  f(x,a(),y,b(),z)

Associative matching

- We consider the flat forms
  - f(a,b,c,d)
  - f(x,c,y)
- and we can write
  - f(x,c,y) << f(a,b,c,d)
  - f(x,d,y) << f(a,b,c,d)

Associative symbols in Tom

```
%gom {
        module ListSet
        imports int
        abstract syntax

         Set = conc(int*)
}

public static void main(String[] args){
        Set s = `conc(1,2,3,4);
        System.out.println("s = "+s);
}
```

Question

Check if the second element of a list of four elements is a “2”

```
%match(s) {
   conc(x,2,y,z) -> { System.out.println("2 found"); }
}
```

Questions

- What are the types of x, y and z in conc(x, 2,y,z)?
  - answer : int
- Fill in the the "..." in conc(...,2,...) to obtain a pattern that matches against conc(1,2,3,4) and conc(1,2,3)?
  - we need variables of type Set

Question

- Check if a list contains a “2”

```
%match(s) {
   conc(X*,2,Y*) -> { System.out.println("2 found"); }
}
```

- Check if the second element of a list is a “2”

```
%match(s) {
   conc(x,2,Y*) -> { System.out.println("2 found"); }
}
```

(old) Exercise

- define a signature to handle sets
- define the operations
  - occurs : E x S -> Bool
  - union : S x S -> S
  - subset, eq : S x S -> Bool
  - diff : S x S -> S

Multisets to Sets

- How to get rid of doubles?
  - conc(X*,y,y,Z*) -> conc(X*,y,Z*)
    - eliminates consecutive doubles
- How to eliminate distant doubles?
  - conc(X*,y, *,y,Z*) -> conc(X*,y,T*,Z*)

Demo

Exercise

- Implement an addressbook with the names and birthdates of a list of persons
  - give the pattern that looks for the persons with the same name
  - give the pattern that finds the persons with a given birthday

Persons

```
%gom{
    module Persons
    imports int String
  abstract syntax

    Date = date(year:int, month:int, day:int)
    Person = person(fn:String,ln:String,bdate:Date)

    PersonList = concPerson( Person* )
}
```

Happy birthday

```
void happyBirthday(PersonList book, Date date) {
  %match(book) {
    concPerson(_*, person(fn, _, date), _*) -> {
      System.out.println("Happy birthday " + `fn);
    }
  }
}
```

Same name

```
%match(book) {
   concPerson(_*, p1@person(fn1,name,_), _*,
                  p2@person(fn2,name,_), _*) -> {
          System.out.println(`p1+" and " +`p2 +
                 "have the same family name");
   }
}
```

Exercise

- Given an order on the elements of the list
- Define a sorting algorithm for the list
- Implement it in Tom

```
public static PersonList sort(PersonList book) {
%match(book) {
  concPerson(X*, p1@person(prenom,name1,_), Y*,
                 p2@person(firstname,name2,_), Z*)
   -> {
    if(`name1.compareTo(`name2) > 0) {
      return sort(`concPerson(X*, p2, Y*, p1, Z*));
    }
  }
}
return book;
}
```

Exercise

- Find the members of the same family that have a different firstname

Different firstname

```
%match(book) {
    concPerson(_*, p1@person(fn1,name,_), _*,
                   p2@person(fn2,name,_), _*)  -> {
        if(`fn1 != `fn2)
          System.out.println(`p1+" and " +`p2 +
              "have different firstnames");
    }
}
```

Different firstname

```
%match(book) {
    concPerson(_*, p1@person(fn,name,_), _*,
                   p2@person(!fn,name,_), _*)  -> {
        System.out.println(`p1+" and " +`p2 +
                "have different firstnames");
    }
}
```

Anti-patterns

- complements used in patterns
- specify what you don't want to match
- use the '!' symbol
- !a => everything that is not a
  - g(!a) << g(b)
  - f(x,!x) << f(a,b)

Cars

```
Vehicle = car(m:Brand,c:Color,t:Color)
        | suv(m:Brand,c:Color,t:Color)

Color = red()
      | blue()
      | green()

Brand = Ford()
      | Renault()
      | BMW()
```

Exercise

- Check if a vehicle is
  - not a SUV
  - not red
  - with different colors in and out

Cars

```
%match(c) {
    !suv(_,_,_)  -> {
            System.out.println("not a SUV");
     }
    car(_,!red(),_)  -> {
            System.out.println("not red");
    }
    car(_,x,!x)  -> {
        System.out.println("different colors");
    }
}
```

Associative lists

- list(*,x,*,x,*) : 2 identical elements
- list(*,x,*,!x,*) : 2 different elements
- !list(*,x,*,x,*) : all different
- !list(*,x,*,!x,*) : all the same

What do we have ?

- efficient data-structures (maximal sharing)
- rewrite rules
- ∅, A, and AU matching (non-linear)
- anti-patterns: !conc(_*,a(),_*)
- everything, smoothly integrated into Java

Hooks

Gom

- define a data structure and
- rewrite rules
- which specify the prefered (normal) forms like
  - sorted lists
  - reduced fractions

Actions executed at construction time

```
module Bool

abstract syntax

Bool = True()
     | False()
     | Not(b:Bool)
     | And(l:Bool,r:Bool)
     | Or(l:Bool,r:Bool)
```

```
module Bool:rules() { ... }

sort Bool:interface() { ComparableTo<Bool> }

sort Bool:block() { ... }

Not:make(x) { ...}
```

Rules

```
module Bool

abstract syntax
Bool = True()
     | False()
     | Not(b:Bool)
     | And(l:Bool,r:Bool)
     | Or(l:Bool,r:Bool)
```

```
module Bool:rules() {
        Or(True(),_) -> True()
        Or(_,True()) -> True()
}

public static void main(String[] args){
        System.out.println(`Or(False(),True()));
}
>True()
```

Actions executed at construction time

```
module Bool

abstract syntax

Bool = True()
     | False()
     | Not(b:Bool)
     | And(l:Bool,r:Bool)
     | Or(l:Bool,r:Bool)
```

```
Not:make(b) {
        %match(Bool b) {
            True()   -> { return `False(); }
            Not(x)   -> { return `x; }
            And(l,r) -> { return `Or(Not(l),Not(r)); }
            Or(l,r)  -> { return
`And(Not(l),Not(r)); }
        }

 System.out.println(`Not(Or(False(),False())))
 >And(True(),True())
 System.out.println(`Not(And(False(),True())));
 >True()
```

Blocks of source code

```
module Bool

abstract syntax

Bool = True()
     | False()
     | Not(b:Bool)
     | And(l:Bool,r:Bool)
     | Or(l:Bool,r:Bool)
```

```
sort Bool:interface(){ Boolable }
sort Bool:block(){
  public boolean toBool(){
    %match(this){
          False() -> {   return false;   }
      True() -> {  return true;   }
      Neg(a) -> {return ! `a.toBool() ;     }
      Or(a,b) -> {return `a.toBool() || `b.toBool();}
      And(a,b) -> {return `a.toBool() && `b.toBool();}
        }
        return false;
}
```

Exercise

- Define the type of sorted lists

Sorted lists

```
%gom {
        module sortedList
imports int
abstract syntax

Integers = sorted(int*)

sorted:make_insert(e,l) {
        %match(l) {
                sorted(head,tail*) -> {
                        if(e >= `head) {
                                return `realMake(head,sorted(e,tail*));
                        }
                }
        }
        return `realMake(e,l);
    }
}
```